/+ dub.sdl:
    name "F"
    dependency "dunkelheit" version=">=0.9.0"
+/

import std.stdio, std.algorithm, std.range, std.conv;
// import dkh.foundation, dkh.scanner;

// import dkh.numeric.convolution;

// import dkh.functional;

// import dkh.modint;
alias Mint = ModInt!(998244353);

Mint[] fact, iFac, p2;
static this() {
    fact = factTable!Mint(10_000);
    iFac = invFactTable!Mint(10_000);
    p2 = new Mint[10000];
    p2[0] = Mint(1);
    foreach (i; 1..10000) {
        p2[i] = p2[i-1] * Mint(2);
    }
}

Mint C(int n, int k) {
    if (n < k || n < 0 || k < 0) return Mint(0);
    return fact[n] * iFac[k] * iFac[n-k];
}

/*memoCont!gBase g;
Mint gBase(int n, int k) {
    Mint ans;
    if (k == 0) return C(n+1, 2);
    return C(n+2, k+2);
}*/

memoCont!fBase f;
Mint fBase(int n, int m) {
    if (n == 0) return Mint(0);
    if (m == 1) {
        return Mint(1);
    }
    Mint ans = f(n, m-1) * C(n+1, 2);
    foreach (k; 1..n+1) {
        ans += f(n-k, m-1) * C(n+2, k+2);
    }

//    writeln("f : ", n, " ", m, " ", ans);
    return ans;
}

int main() {
    Scanner sc = new Scanner(stdin);
    scope(exit) assert(!sc.hasNext);
    int n, m;
    sc.read(n, m);
/*    if (300 < n) {
        writeln("TODO DESI???");
        return 1;
    }*/
    
//    g.init([[0, n], [0, n]]);
//    f.init([[0, n], [0, m]]);

    Mint[] pol = new Mint[n+1];
    pol[] = Mint(1);
    pol[0] = Mint(0);

    Mint ans = Mint(1);
    foreach (j; 1..m+1) {
        foreach (i; 1..n+1) {
//            assert(f(i, j) == pol[i]);
//            ans += C(n, n-i) * C(m, m-j) * f(i, j); 
            ans += C(n, n-i) * C(m, m-j) * pol[i];
        }
        foreach (i; 0..n+1) pol[i] *= iFac[i];
        Mint[] le = new Mint[n+1];
        foreach (i; 1..n+1) le[i] = iFac[i+2];
        Mint[] npol = multiply!(3, Mint)(pol, le);
        npol.length = n+1;
        foreach (i; 0..n+1) {
            npol[i] *= fact[i+2];
            npol[i] += pol[i] * fact[i] * C(i+1, 2);
        }
        pol = npol;
    }

    writeln(ans);

    return 0;
}
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/ldc/inline.d */
// module dkh.ldc.inline;

version(LDC) {
    pragma(LDC_inline_ir) R inlineIR(string s, R, P...)(P);
}
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/container/stackpayload.d */
// module dkh.container.stackpayload;

 
struct StackPayload(T, size_t MINCAP = 4) if (MINCAP >= 1) {
    import core.exception : RangeError;

    private T* _data;
    private uint len, cap;

    @property bool empty() const { return len == 0; }
    @property size_t length() const { return len; }
    alias opDollar = length;

     
    inout(T)[] data() inout { return (_data) ? _data[0..len] : null; }
    
    ref inout(T) opIndex(size_t i) inout {
        version(assert) if (len <= i) throw new RangeError();
        return _data[i];
    }  
    ref inout(T) front() inout { return this[0]; }  
    ref inout(T) back() inout { return this[$-1]; }  

    void reserve(size_t newCap) {
        import core.memory : GC;
        import core.stdc.string : memcpy;
        import std.conv : to;
        if (newCap <= cap) return;
        void* newData = GC.malloc(newCap * T.sizeof);
        cap = newCap.to!uint;
        if (len) memcpy(newData, _data, len * T.sizeof);
        _data = cast(T*)(newData);
    }  
    void free() {
        import core.memory : GC;
        GC.free(_data);
    }  
     
    void clear() {
        len = 0;
    }

    void insertBack(T item) {
        import std.algorithm : max;
        if (len == cap) reserve(max(cap * 2, MINCAP));
        _data[len++] = item;
    }  
    alias opOpAssign(string op : "~") = insertBack;  
    void removeBack() {
        assert(!empty, "StackPayload.removeBack: Stack is empty");
        len--;
    }  
}

 
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/functional.d */
// module dkh.functional;

 
struct memoCont(alias pred) {
    import std.traits : ReturnType, ParameterTypeTuple, isIntegral;
    import std.meta : allSatisfy;
    alias R = ReturnType!pred;
    alias Args = ParameterTypeTuple!pred;
    static assert (allSatisfy!(isIntegral, Args));
    static immutable N = Args.length;
    
    private int[2][N] rng;
    int[N] len;
    R[] dp;
    bool[] used;
    void init(in int[2][N] rng) {
        import std.algorithm : reduce, map;
        import std.range : array;
        this.rng = rng;
        len = rng[].map!(a => a[1]-a[0]+1).array;
        auto sz = reduce!"a*b"(1, len);
        dp = new R[sz];
        used = new bool[sz];
    }
    R opCall(Args args) {
        import core.exception : RangeError;
        size_t idx, base = 1;
        foreach (i, v; args) {
            version(assert) {
                if (v < rng[i][0] || rng[i][1] < v) {
                    throw new RangeError;
                }
            }
            assert(rng[i][0] <= v && v <= rng[i][1]);
            idx += base*(v - rng[i][0]);
            base *= len[i];
        }
        if (used[idx]) return dp[idx];
        used[idx] = true;
        auto r = pred(args);
        dp[idx] = r;
        return r;
    }
}

 
 
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/foundation.d */
 
// module dkh.foundation;

 
static if (__VERSION__ <= 2070) {
    /*
    Copied by https://github.com/dlang/phobos/blob/master/std/algorithm/iteration.d
    Copyright: Andrei Alexandrescu 2008-.
    License: $(HTTP boost.org/LICENSE_1_0.txt, Boost License 1.0).
    */
    template fold(fun...) if (fun.length >= 1) {
        auto fold(R, S...)(R r, S seed) {
            import std.algorithm : reduce;
            static if (S.length < 2) {
                return reduce!fun(seed, r);
            } else {
                import std.typecons : tuple;
                return reduce!fun(tuple(seed), r);
            }
        }
    }
     
}
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/modint.d */
// module dkh.modint;

// import dkh.numeric.primitive;

 
struct ModInt(uint MD) if (MD < int.max) {
    import std.conv : to;
    uint v;
    this(int v) {this(long(v));}
    this(long v) {this.v = (v%MD+MD)%MD;}
    static auto normS(uint x) {return (x<MD)?x:x-MD;}
    static auto make(uint x) {ModInt m; m.v = x; return m;}
     
    auto opBinary(string op:"+")(ModInt r) const {return make(normS(v+r.v));}
     
    auto opBinary(string op:"-")(ModInt r) const {return make(normS(v+MD-r.v));}
     
    auto opBinary(string op:"*")(ModInt r) const {return make((ulong(v)*r.v%MD).to!uint);}
     
    auto opBinary(string op:"/")(ModInt r) const {return this*inv(r);}
    auto opOpAssign(string op)(ModInt r) {return mixin ("this=this"~op~"r");}
     
    static ModInt inv(ModInt x) {return ModInt(extGcd!int(x.v, MD)[0]);}
    string toString() const {return v.to!string;}
}

 
 

 

 
struct DModInt(string name) {
    import std.conv : to;
    static uint MD;
    uint v;
    this(int v) {this(long(v));}
    this(long v) {this.v = ((v%MD+MD)%MD).to!uint;}
    static auto normS(uint x) {return (x<MD)?x:x-MD;}
    static auto make(uint x) {DModInt m; m.MD = MD; m.v = x; return m;}
     
    auto opBinary(string op:"+")(DModInt r) const {return make(normS(v+r.v));}
     
    auto opBinary(string op:"-")(DModInt r) const {return make(normS(v+MD-r.v));}
     
    auto opBinary(string op:"*")(DModInt r) const {return make((ulong(v)*r.v%MD).to!uint);}
     
    auto opBinary(string op:"/")(DModInt r) const {return this*inv(r);}
    auto opOpAssign(string op)(DModInt r) {return mixin ("this=this"~op~"r");}
     
    static DModInt inv(DModInt x) {
        return DModInt(extGcd!int(x.v, MD)[0]);
    }
    string toString() {return v.to!string;}
}

 
 

 

template isModInt(T) {
    const isModInt =
        is(T : ModInt!MD, uint MD) || is(S : DModInt!S, string s);
}


T[] factTable(T)(size_t length) if (isModInt!T) {
    import std.range : take, recurrence;
    import std.array : array;
    return T(1).recurrence!((a, n) => a[n-1]*T(n)).take(length).array;
}

T[] invFactTable(T)(size_t length) if (isModInt!T) {
    import std.algorithm : map, reduce;
    import std.range : take, recurrence, iota;
    import std.array : array;
    auto res = new T[length];
    res[$-1] = T(1) / iota(1, length).map!T.reduce!"a*b";
    foreach_reverse (i, v; res[0..$-1]) {
        res[i] = res[i+1] * T(i+1);
    }
    return res;
}

T[] invTable(T)(size_t length) if (isModInt!T) {
    auto f = factTable!T(length);
    auto invf = invFactTable!T(length);
    auto res = new T[length];
    foreach (i; 1..length) {
        res[i] = invf[i] * f[i-1];
    }
    return res;
}

 
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/scanner.d */
// module dkh.scanner;

// import dkh.container.stackpayload;

 
class Scanner {
    import std.stdio : File;
    import std.conv : to;
    import std.range : front, popFront, array, ElementType;
    import std.array : split;
    import std.traits : isSomeChar, isStaticArray, isArray; 
    import std.algorithm : map;
    File f;
    this(File f) {
        this.f = f;
    }
    char[512] lineBuf;
    char[] line;
    private bool succW() {
        import std.range.primitives : empty, front, popFront;
        import std.ascii : isWhite;
        while (!line.empty && line.front.isWhite) {
            line.popFront;
        }
        return !line.empty;
    }
    private bool succ() {
        import std.range.primitives : empty, front, popFront;
        import std.ascii : isWhite;
        while (true) {
            while (!line.empty && line.front.isWhite) {
                line.popFront;
            }
            if (!line.empty) break;
            line = lineBuf[];
            f.readln(line);
            if (!line.length) return false;
        }
        return true;
    }

    private bool readSingle(T)(ref T x) {
        import std.algorithm : findSplitBefore;
        import std.string : strip;
        import std.conv : parse;
        if (!succ()) return false;
        static if (isArray!T) {
            alias E = ElementType!T;
            static if (isSomeChar!E) {
                 
                 
                auto r = line.findSplitBefore(" ");
                x = r[0].strip.dup;
                line = r[1];
            } else static if (isStaticArray!T) {
                foreach (i; 0..T.length) {
                    bool f = succW();
                    assert(f);
                    x[i] = line.parse!E;
                }
            } else {
                StackPayload!E buf;
                while (succW()) {
                    buf ~= line.parse!E;
                }
                x = buf.data;
            }
        } else {
            x = line.parse!T;
        }
        return true;
    }

    int unsafeRead(T, Args...)(ref T x, auto ref Args args) {
        if (!readSingle(x)) return 0;
        static if (args.length == 0) {
            return 1;
        } else {
            return 1 + read(args);
        }
    }
    void read(Args...)(auto ref Args args) {
        import std.exception;
        static if (args.length != 0) {
            enforce(readSingle(args[0]));
            read(args[1..$]);
        }
    }
    bool hasNext() {
        return succ();
    }
}


 
 

 
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/numeric/primitive.d */
// module dkh.numeric.primitive;

import std.traits;
import std.bigint;

 
Unqual!T pow(T, U)(T x, U n)
if (!isFloatingPoint!T && (isIntegral!U || is(U == BigInt))) {
    return pow(x, n, T(1));
}

 
Unqual!T pow(T, U, V)(T x, U n, V e)
if ((isIntegral!U || is(U == BigInt)) && is(Unqual!T == Unqual!V)) {
    Unqual!T b = x, v = e;
    Unqual!U m = n;
    while (m) {
        if (m & 1) v *= b;
        b *= b;
        m /= 2;
    }
    return v;
}

 

 
T powMod(T, U, V)(T x, U n, V md)
if (isIntegral!U || is(U == BigInt)) {
    T r = T(1);
    Unqual!U m = n;
    while (m) {
        if (m & 1) r = (r*x)%md;
        x = (x*x)%md;
        m >>= 1;
    }
    return r % md;
}

 

// import dkh.int128;

 
ulong ulongPowMod(U)(ulong x, U n, ulong md)
if (isIntegral!U || is(U == BigInt)) {
    x %= md;
    ulong r = 1;
    while (n) {
        if (n & 1) {
            r = mul128(r, x).mod128(md);
        }
        x = mul128(x, x).mod128(md);
        n >>= 1;
    }
    return r % md;
}

 
T lcm(T)(in T a, in T b) {
    import std.numeric : gcd;
    return a / gcd(a,b) * b;
}

 
 

 
 
T[3] extGcd(T)(in T a, in T b) 
if (!isIntegral!T || isSigned!T)  
{
    if (b==0) {
        return [T(1), T(0), a];
    } else {
        auto e = extGcd(b, a%b);
        return [e[1], e[0]-a/b*e[1], e[2]];
    }
}

 
 

 
T invMod(T)(T x, T md) {
    auto r = extGcd!T(x, md);
    assert(r[2] == 1);
    auto z = r[0];
    return (z % md + md) % md;
}

 
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/int128.d */
 

// module dkh.int128;

version(LDC) {
//     import dkh.ldc.inline;
}

version(LDC) version(X86_64) {
    version = LDC_IR;
}

 
ulong[2] mul128(ulong a, ulong b) {
    ulong[2] res;
    version(LDC_IR) {
        ulong upper, lower;
        inlineIR!(`
            %r0 = zext i64 %0 to i128 
            %r1 = zext i64 %1 to i128
            %r2 = mul i128 %r1, %r0
            %r3 = trunc i128 %r2 to i64
            %r4 = lshr i128 %r2, 64
            %r5 = trunc i128 %r4 to i64
            store i64 %r3, i64* %2
            store i64 %r5, i64* %3`, void)(a, b, &lower, &upper);
        return [lower, upper];
    } else version(D_InlineAsm_X86_64) {
        ulong upper, lower;
        asm {
            mov RAX, a;
            mul b;
            mov lower, RAX;
            mov upper, RDX;
        }
        return [lower, upper];
    } else {
        ulong B = 2UL^^32;
        ulong[2] a2 = [a % B, a / B];
        ulong[2] b2 = [b % B, b / B];
        ulong[4] c;
        foreach (i; 0..2) {
            foreach (j; 0..2) {
                c[i+j] += a2[i] * b2[j] % B;
                c[i+j+1] += a2[i] * b2[j] / B;
            }
        }
        foreach (i; 0..3) {
            c[i+1] += c[i] / B;
            c[i] %= B;
        }
        return [c[0] + c[1] * B, c[2] + c[3] * B];
    }
}

 

 
ulong div128(ulong[2] a, ulong b) {
    version(LDC_IR) {
        return inlineIR!(`
            %r0 = zext i64 %0 to i128
            %r1 = zext i64 %1 to i128
            %r2 = shl i128 %r1, 64
            %r3 = add i128 %r0, %r2
            %r4 = zext i64 %2 to i128
            %r5 = udiv i128 %r3, %r4
            %r6 = trunc i128 %r5 to i64
            ret i64 %r6`,ulong)(a[0], a[1], b);
    } else version(D_InlineAsm_X86_64) {
        ulong upper = a[1], lower = a[0];
        ulong res;
        asm {
            mov RDX, upper;
            mov RAX, lower;
            div b;
            mov res, RAX;
        }
        return res;
    } else {
        if (b == 1) return a[0];
        while (!(b & (1UL << 63))) {
            a[1] <<= 1;
            if (a[0] & (1UL << 63)) a[1] |= 1;
            a[0] <<= 1;
            b <<= 1;
        }
        ulong ans = 0;
        foreach (i; 0..64) {
            bool up = (a[1] & (1UL << 63)) != 0;
            a[1] <<= 1;
            if (a[0] & (1UL << 63)) a[1] |= 1;
            a[0] <<= 1;

            ans <<= 1;
            if (up || b <= a[1]) {
                a[1] -= b;
                ans++;
            }
        }
        return ans;
    }
}


 
ulong mod128(ulong[2] a, ulong b) {
    version(D_InlineAsm_X86_64) {
        ulong upper = a[1], lower = a[0];
        ulong res;
        asm {
            mov RDX, upper;
            mov RAX, lower;
            div b;
            mov res, RDX;
        }
        return res;
    } else {
        return a[0] - div128(a, b) * b;
    }
}

 
/* IMPORT /Users/yosupo/Program/dunkelheit/source/dkh/numeric/convolution.d */
// module dkh.numeric.convolution;

 
T[] zeta(T)(T[] v, bool rev) {
    import core.bitop : bsr;
    int n = bsr(v.length);
    assert(1<<n == v.length);
    foreach (fe; 0..n) {
        foreach (i, _; v) {
            if (i & (1<<fe)) continue;
            if (!rev) {
                v[i] += v[i|(1<<fe)];
            } else {
                v[i] -= v[i|(1<<fe)];
            }
        }
    }
    return v;
}

 
T[] hadamard(T)(T[] v, bool rev) {
    import core.bitop : bsr;
    int n = bsr(v.length);
    assert(1<<n == v.length);
    foreach (fe; 0..n) {
        foreach (i, _; v) {
            if (i & (1<<fe)) continue;
            auto l = v[i], r = v[i|(1<<fe)];
            if (!rev) {
                v[i] = l+r;
                v[i|(1<<fe)] = l-r;
            } else {
                v[i] = (l+r)/2;
                v[i|(1<<fe)] = (l-r)/2;
            }
        }
    }
    return v;
}

import std.complex;

double[] fftSinList(size_t S) {
    import std.math : PI, sin;
    assert(2 <= S);
    size_t N = 1<<S;
    static double[][30] buf;
    if (!buf[S].length) {
        buf[S] = new double[3*N/4+1];
        foreach (i; 0..N/4+1) {
            buf[S][i] = sin(i*2*double(PI)/N);
            buf[S][N/2-i] = buf[S][i];
            buf[S][N/2+i] = -buf[S][i];
        }
    }
    return buf[S];
}

 
void fft(bool type)(Complex!double[] c) {
    import std.algorithm : swap;
    import core.bitop : bsr;
    alias P = Complex!double;
    size_t N = c.length;
    assert(N);
    size_t S = bsr(N);
    assert(1<<S == N);
    if (S == 1) {
        auto x = c[0], y = c[1];
        c[0] = x+y;
        c[1] = x-y;
        return;
    }
    auto rot = fftSinList(S);
    P[] a = c.dup, b = new P[c.length];
    foreach (i; 1..S+1) {
        size_t W = 1<<(S-i);
        for (size_t y = 0; y < N/2; y += W) {
            P now = P(rot[y + N/4], rot[y]);
            if (type) now = conj(now);
            foreach (x; 0..W) {
                auto l =       a[y<<1 | x];
                auto r = now * a[y<<1 | x | W];
                b[y | x]        = l+r;
                b[y | x | N>>1] = l-r;
            }
        }
        swap(a, b);
    }
    c[] = a[];
}

 
double[] multiply(in double[] a, in double[] b) {
    alias P = Complex!double;
    size_t A = a.length, B = b.length;
    if (!A || !B) return [];
    size_t lg = 1;
    while ((1<<lg) < A+B-1) lg++;
    size_t N = 1<<lg;
    P[] d = new P[N];
    d[] = P(0, 0);
    foreach (i; 0..A) d[i].re = a[i];
    foreach (i; 0..B) d[i].im = b[i];
    fft!false(d);
    foreach (i; 0..N/2+1) {
        auto j = i ? (N-i) : 0;
        P x = P(d[i].re+d[j].re, d[i].im-d[j].im);
        P y = P(d[i].im+d[j].im, -d[i].re+d[j].re);
        d[i] = x * y / 4;
        if (i != j) d[j] = conj(d[i]);
    }
    fft!true(d);
    double[] c = new double[A+B-1];
    foreach (i; 0..A+B-1) {
        c[i] = d[i].re / N;
    }
    return c;
}

 

// import dkh.modint, dkh.numeric.primitive;

void nft(uint G, bool type, Mint)(Mint[] c) {
    import std.algorithm : swap;
    import core.bitop : bsr;    
    size_t N = c.length;
    assert(N);
    size_t S = bsr(N);
    assert(1<<S == N);

    Mint[] a = c.dup, b = new Mint[N];
    foreach (i; 1..S+1) {
        size_t W = 1<<(S-i);
        Mint base = pow(Mint(G), Mint(-1).v/(1<<i));
        if (type) base = Mint(1)/base;
        Mint now = 1;
        for (size_t y = 0; y < N/2; y += W) {
            foreach (x; 0..W) {
                auto l =       a[y<<1 | x];
                auto r = now * a[y<<1 | x | W];
                b[y | x]        = l+r;
                b[y | x | N>>1] = l-r;
            }
            now *= base;
        }
        swap(a, b);
    }
    c[] = a[];    
}

Mint[] multiply(uint G, Mint)(in Mint[] a, in Mint[] b) {
    size_t A = a.length, B = b.length;
    if (!A || !B) return [];
    size_t lg = 1;
    while ((1<<lg) < A+B-1) lg++;
    size_t N = 1<<lg;
    Mint[] _a = new Mint[N];
    Mint[] _b = new Mint[N];
    foreach (i; 0..A) _a[i] = a[i];
    foreach (i; 0..B) _b[i] = b[i];
    nft!(G, false)(_a);
    nft!(G, false)(_b);
    foreach (i; 0..N) _a[i] *= _b[i];
    nft!(G, true)(_a);
    Mint[] c = new Mint[A+B-1];
    Mint iN = Mint(1) / Mint(N);
    foreach (i; 0..A+B-1) {
        c[i] = _a[i] * iN;
    }
    return c;
}

Mint[] multiply(Mint, size_t M = 3, size_t W = 10)(in Mint[] a, in Mint[] b)
if (isModInt!Mint) {
    import std.math : round;
    alias P = Complex!double;

    size_t A = a.length, B = b.length;
    if (!A || !B) return [];
    auto N = A + B - 1;
    size_t lg = 1;
    while ((1<<lg) < N) lg++;
    N = 1<<lg;

    P[][M] x, y;
    P[] w = new P[N];
    foreach (ph; 0..M) {
        x[ph] = new P[N];
        y[ph] = new P[N];
        w[] = P(0, 0);
        foreach (i; 0..A) w[i].re = (a[i].v >> (ph*W)) % (1<<W);
        foreach (i; 0..B) w[i].im = (b[i].v >> (ph*W)) % (1<<W);
        fft!false(w);
        foreach (i; 0..N) w[i] *= 0.5;
        foreach (i; 0..N) {
            auto j = i ? N-i : 0;
            x[ph][i] = P(w[i].re+w[j].re,  w[i].im-w[j].im);
            y[ph][i] = P(w[i].im+w[j].im, -w[i].re+w[j].re);
        }
    }

    Mint[] c = new Mint[A+B-1];
    Mint basel = 1, baser = pow(Mint(1<<W), M);
    P[] z = new P[N];
    foreach (ph; 0..M) {
        z[] = P(0, 0);
        foreach (af; 0..ph+1) {
            auto bf = ph - af;
            foreach (i; 0..N) {
                z[i] += x[af][i] * y[bf][i];
            }
        }
        foreach (af; ph+1..M) {
            auto bf = ph + M - af;
            foreach (i; 0..N) {
                z[i] += x[af][i] * y[bf][i] * P(0, 1);
            }
        }
        fft!true(z);
        foreach (i; 0..A+B-1) {
            z[i] *= 1.0/N;
            c[i] += Mint(cast(long)(round(z[i].re)))*basel;
            c[i] += Mint(cast(long)(round(z[i].im)))*baser;
        }
        basel *= Mint(1<<W);
        baser *= Mint(1<<W);
    }
    return c;
}

 


 


/*
This source code generated by dunkelheit and include dunkelheit's source code.
dunkelheit's Copyright: Copyright (c) 2016- Kohei Morita. (https://github.com/yosupo06/dunkelheit)
dunkelheit's License: MIT License(https://github.com/yosupo06/dunkelheit/blob/master/LICENSE.txt)
*/


Measured memory_bytes: 4732